/*
Andrea Headley	
Proposal title: Affective Architecture: Isolating the Influence of Physical Environment on Perceptual and Behavioral Attitudes toward Police


HYPOTHESES

Stated-Hyp1: "we hypothesize that “welcoming” architectural designs will improve perceptions of police stations, leading to increased positive affect towards police and thus more cooperative behavioral intentions (e.g., crime reporting)." (p. 2)

"we hypothesize that “hostile/defensive” police buildings will engender more negative perceptions toward, and lead to, reduced interactions with police" (p. 2)

	Test-Hyp1: Professionalism perception will be higher for welcoming buildings vs. hostile buildings.

	Test-Hyp2: Positive affect will be higher for approaching welcoming buildings vs. hostile buildings.	
	
	Test-Hyp3: Liklihood of entering will be higher for welcoming buildings vs. hostile buildings.	


********************************************************************************
NOTES:

- there is no control group (i.e. there's only welcoming and hostile - two groups). 

- there are a bunch of post-treatment measures that the proposal says are 
exploratory (p.7). We regard these as phantom outcomes.

- there are 3 main outcomes on page 6: feelings approaching the building, 
professionalism perception, liklihood of entering.
*/


clear all
use "Headley1126.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATIONS
	tab P_HEADLEY
	recode P_HEADLEY (1/3=1) (4/6=2), gen(buildingtype)
	tab buildingtype	
	lab def build 1 "1 hostile" 2 "2 welcoming"
	lab val buildingtype build
	
* OUTCOME MEASURES

* positive affect
		/* 
	a	[Safe as I am approaching this building] I would feel… -- reverse code
	b	[Confident as I am approaching this building] I would feel… -- reverse code
	c	[Jittery as I am approaching this building] I would feel…
	d	[Optimistic as I am approaching this building] I would feel… -- reverse code
	e	[Nervous as I am approaching this building] I would feel…
	f	[Calm as I am approaching this building] I would feel… -- reverse code
	g	[Worried as I am approaching this building] I would feel…
	h	[Content as I am approaching this building] I would feel… -- reverse code
	i	[Relaxed as I am approaching this building] I would feel… -- reverse code
		
		responses:
	1	Clearly describes my feelings
	2	Mostly describes my feelings
	3	Moderately describes my feelings
	4	Slightly describes my feelings
	5	Does not describe my feelings
		*/

	foreach var in A B C D E F G H I {
		replace Q1`var'=. if Q1`var'>5
		tab Q1`var'
	}
		
	* recode reverse coded items: higher values reflect positive feelings	
	foreach var in A B D F H I {
		replace Q1`var'= 6-Q1`var'
		tab Q1`var'
	}
	
	* create scale
	gen positiveaff= Q1A + Q1B+Q1C +Q1D +Q1E+Q1F+Q1G+Q1H+Q1I
	tab positiveaff
	
* liklihood of entering
	// How likely are you to proceed in entering the police department?
	replace Q3=. if Q3>5 // recode missing
	gen probenter=6-Q3 // correct reverse code
	tab probenter

* perception of professionalism 	
	/* 
	Q2A	[I feel confident that the criminal activity I am going to report to the police today will be taken seriously.] Please indicate your level of agreement with the following statement:

	Q2B	[The police officers who work here will be professional.] Please indicate your level of agreement with the following statement:
	*/
	
	* recode missing
	foreach var in A B {
		replace Q2`var'=. if Q2`var'>5
		tab Q2`var'
	}
		
	* recode reverse coded items: higher values reflect better perception	
	foreach var in A B {
		replace Q2`var'= 6-Q2`var'
		tab Q2`var'
	}
	
	* create scale
	gen professionalism= Q2A + Q2B
	tab professionalism

	
********************************************************************************

* ANALYSIS
	
*Test-Hyp1: Professionalism perception will be higher for welcoming buildings vs. hostile buildings.
	reg professionalism i.buildingtype
	// reject. 0.394
	tess 2.buildingtype +, init(Headley1126) bonf(3)
	
*Test-Hyp2: Positive affect will be higher for approaching welcoming buildings vs. hostile buildings.	
	reg positiveaff i.buildingtype
	// reject. 0.195
	tess 2.buildingtype +, bonf(3)
	
*Test-Hyp3: Liklihood of entering will be higher for welcoming buildings vs. hostile buildings.	
	reg probenter i.buildingtype
	// reject. 0.275
	tess 2.buildingtype +, bonf(3)
